ASSIGNMENT-1

-D.Lakshmi Sivani
EE18BTECH11012

1) a)Operating point through ngspice netlist file is

Simulated file and output:



File: /home/sivani/A1/Q1a.net Page 1 of 1

```
*Netlist
Vin 1 0 2.5V
R   1 out 1k
Rs  3 0 100
C   out 3 1u

*Control commands
.control
op
print V(out)
```




```
ngspice 1 -> source Q1a.net

Circuit: *netlist

Doing analysis at TEMP = 27.000000 and TNOM = 27.000000



No. of Data Rows : 1
v(out) = 2.500000e+00
ngspice 1 ->
```

So, Vout = 2.5 V

1) b) Given Vin=sinwt and w=100 Hz and w=1M Hz
So for the given circuit the phase lags in both case-1(when w=100 Hz) and case-2(when w=1M Hz) are solved as:

1) b) $V_{in} = \sin\omega t$



Case-1: When $\omega = 100\,Hz\,(f)$

$$\frac{V_{out}}{V_{in}} = \frac{i(1/j\omega C + R_s)}{i(1/j\omega C + (R+R_s))}$$

$$\therefore \frac{V_{out}}{V_{in}} = \frac{1 + j\omega C(R_s)}{1 + j\omega C(R+R_s)}$$

$$\text{Phase-lag}(\phi) = \tan^{-1}\left(\frac{R_s\omega C}{1}\right) - \tan^{-1}\left(\frac{(R+R_s)\omega C}{1}\right)$$

$$\therefore \boxed{\phi = \tan^{-1}(R_s\omega C) - \tan^{-1}((R+R_s)\omega C)} \quad \text{—(1)}$$

$$\therefore \phi = \tan^{-1}(10^2 \times 2\pi \times 10^2 \times 10^{-6}) - \tan^{-1}(10^2 \times 2\pi \times 10^2 \times 11 \times 10^{-6})$$

$$\therefore \phi = \tan^{-1}(2\pi \times 10^{-2}) - \tan^{-1}(2\pi \times 11 \times 10^{-2})$$

$$\therefore \phi = 3.595^\circ - 34.6503^\circ \simeq \underline{31.05^\circ}$$

Case-2: When $\omega = 1MHz(f) \Rightarrow \omega = 2\pi \times 10^6\ rad/s$

From eq-(1),

$$\phi = \tan^{-1}(R_s\omega C) - \tan^{-1}((R+R_s)\omega C)$$

$$\therefore \phi = \tan^{-1}(2\pi \times 10^2) - \tan^{-1}(2\pi \times 11 \times 10^2)$$

$$= 89.908^\circ - 89.9917^\circ$$

$$\simeq \underline{0.0837^\circ}$$

Simulation results:

In case-1: When w = 100 Hz

Netlist file:

```
Open ▾                                   Save  ≡
Q1b_1.net ×   Q1c.net ×   Q3.net ×   Q2b.net ×   Q4a_1.net ×

*Title:Transient analysis

*Netlist
Vin 1 0 Dc 0 AC 1 SIN(0V 1V 100Hz 0 0 0)
R   1 out 1k
Rs  3 0 100
C   out 3 1u


*Analysis for 50ms,step size 0.01ms

.tran 0.01ms 50ms

*Control commands
.control
op
run
plot V(1) V(out)

meas TRAN T1 TRIG V(1) VAL=0 CROSS=2 TARG V(1) VAL=0 CROSS=4
meas TRAN T2 TRIG V(1) VAL=0 CROSS=2 TARG V(out) VAL=0 CROSS=2
let Lag = -360*(T2/T1)
print Lag


.endc
.end
```




```
sivani@sivani-Inspiron-5558: ~/A1
File Edit View Search Terminal Help
No. of Data Rows : 1
Doing analysis at TEMP = 27.000000 and TNOM = 27.000000


Initial Transient Solution
--------------------------

Node                                   Voltage
----                                   -------
1                                            0
out                                          0
3                                            0
vin#branch                                   0

 Reference value :  2.49973e-02

No. of Data Rows : 5012
t1                  =  1.000000e-02 targ=  1.500000e-02 trig=  5
.000000e-03
t2                  =  8.666058e-04 targ=  5.866606e-03 trig=  5
.000000e-03
lag = -3.11978e+01
ngspice 1 ->
```

So, The phase-lag obtained in both analytical approach and simulation are same(which is approximately equal to 31.1 degrees).

In case-2: When w = 1 M Hz

Netlist file:

```
*Title:Transient analysis

*Netlist
Vin 1 0 Dc 0 AC 1 SIN(0V 1V 1e6Hz 0 0 0)
R   1 out 1k
Rs  3 0 100
C   out 3 1u


*Analysis for 5us,step size 2ns

.tran 2n 5us

*Control commands
.control
op
run
plot V(1) V(out)

meas TRAN T1 TRIG V(1) VAL=0 CROSS=2 TARG V(1) VAL=0 CROSS=4
meas TRAN T2 TRIG V(1) VAL=0 CROSS=2 TARG V(out) VAL=0 CROSS=2
let Lag = -360*(T2/T1)
print Lag


.endc
.end
```




```
No. of Data Rows : 1
Doing analysis at TEMP = 27.000000 and TNOM = 27.000000


Initial Transient Solution
--------------------------

Node                                    Voltage
----                                    -------
1                                             0
out                                           0
3                                             0
vin#branch                                    0


No. of Data Rows : 2508
t1                  =  1.000000e-06 targ=  1.500000e-06 trig=  5
.000000e-07
t2                  =  4.604429e-10 targ=  5.004604e-07 trig=  5
.000000e-07
lag = -1.65759e-01
ngspice 1 ->
```

Here also,The phase-lag obtained in both analytical approach and simulation are approximately around 0 degrees.

1) c) Amplitude and Phase transfer characteristics are observed in this way using ngspice simulation:

```
*Title:AC analysis

*Netlist
Vin 1 0 Dc 0 AC 1 SIN(0V 1V 100Hz 0 0 0)
R   1 out 1k
Rs  3 0 100
C   out 3 1u

*AC analysis for 1 Hz to 1MHz, 10 points per decade
.ac dec 10 1 1Meg
.control
run

*Magnitude dB plot for v(out) on log scale
plot vdb(out) xlog
*Phase degrees plot for v(out) on log scale
plot {57.29*vp(out)} xlog
meas AC f_3db FIND frequency WHEN Vdb(out)=-3db
let phase = {57.29*vp(out)}
meas AC Phase_lag FIND phase WHEN frequency = f_3db
.endc
.end
```

Plain Text   Tab Width: 8   Ln 24, Col 1   INS

Amplitude Characteristics:



Phase tranfer Characteristics:





```
sivani@sivani-Inspiron-5558: ~/A1
File  Edit  View  Search  Terminal  Help
out                                                     0
3                                                       0
vin#branch                                              0


No. of Data Rows : 2508
t1                  =  1.000000e-06 targ=  1.500000e-06 trig=  5
.000000e-07
t2                  =  4.604429e-10 targ=  5.004604e-07 trig=  5
.000000e-07
lag = -1.65759e-01
ngspice 1 -> source Q1c.net

Circuit: *title:ac analysis

Doing analysis at TEMP = 27.000000 and TNOM = 27.000000


No. of Data Rows : 61
f_3db               =  1.455510e+02
phase_lag           =  -3.976622e+01
ngspice 1 ->
```

From the simulation results,we got 3db frequency = 145.55 Hz and Phase lag =39.76 degrees.

When we calculated these values in analytical approach then

1) c) $v_{in} = \sin \omega t$



Here Transfer function is

$$\frac{v_o}{v_{in}} = \frac{1 + j\omega C R_s}{1 + j\omega C(R + R_s)}$$

At 3dB, $\left|\frac{v_o}{v_{in}}\right| = \frac{1}{\sqrt{2}}$

$$\Rightarrow \frac{\sqrt{1 + \omega^2 C^2 R_s^2}}{\sqrt{1 + \omega^2 C^2 (R + R_s)^2}} = \frac{1}{\sqrt{2}}$$

(Assumed $\pi^2 \approx 10$)

$$\Rightarrow \frac{\sqrt{1 + \omega^2 (10^{-6} \times 10^2)^2}}{\sqrt{1 + \omega^2 (11 \times 10^2 \times 10^{-6})^2}} = \frac{1}{\sqrt{2}}$$

$$\Rightarrow \frac{\sqrt{10^8 + \omega^2}}{\sqrt{10^8 + 121\omega^2}} = \frac{1}{\sqrt{2}}$$

$$\Rightarrow 2 \times 10^8 + 2\omega^2 = 10^8 + 121\omega^2$$

$$\Rightarrow 119\omega^2 = 10^8$$

$$\therefore \omega = \frac{10^4}{\sqrt{119}} \Rightarrow f = \frac{10^4}{119} \times \frac{1}{2\pi} \text{ Hz}$$

$$\therefore \boxed{f \approx 145.5 \text{ Hz}}$$

At $f = 145.5$ Hz phase-lag is (i.e, at 3dB)

$$\phi = \angle H(j\omega)_{3dB} = \tan^{-1}\left(\frac{2\pi \times 145.5}{10^4}\right) - \tan^{-1}\left(\frac{2\pi \times 145.5 \times 11}{10^4}\right)$$

$$\therefore \boxed{\phi \simeq -39.76^\circ}$$

Even in analytical approach we got 3db frequency = 145.55 Hz and Phase lag =39.76 degrees.
So,our results in both approaches are accurate.

2) a) Analytical Calculation for soliving Id is

2) a) Analytic Calculation:-

Applying KVL for the given circuit then



$$2.5 - (2K)I_D - 0.7 - (2K)I_D - 0.7 = 0$$

$$\Rightarrow (4K)I_D + 1.4 = 2.5$$

$$\Rightarrow (4K)I_D = 1.1$$

$$\therefore \boxed{I_D = 0.275\,mA}$$

From analytical calculation Id = 0.275 mA.

2) b) Analytical Calculation for solving Id and Vd

2) b) Analytic Calculation:-

Apply KVL,

$$(2.5) - (4K)I_D - 2V_D = 0$$

$$(4K)I_D + 2V_D - 2.5 = 0 \quad \text{—(1)}$$

w.k.t,

$$I_D = I_S\left(e^{\frac{qV_D}{nKT}} - 1\right)$$

Given $I_S = 10^{-14}$ A, $T = 300K$

$$\therefore I_D = 10^{-14}\left(e^{\frac{V_D}{0.0259}} - 1\right)$$

$\Rightarrow \frac{KT}{q} = 0.0259\,eV$

Substitute this in eq-(1)

$$\Rightarrow (4K)\times 10^{-14}\left(e^{V_D/0.0259} - 1\right) + 2V_D - 2.5 = 0$$

$$\therefore (4\times 10^{-11})\left(e^{V_D/0.0259} - 1\right) + 2V_D - 2.5 = 0$$

Let $f(x) = (4\times 10^{-11})\left(e^{x/0.0259} - 1\right) + 2x - 2.5 = 0$

Solve this equation by any method. For suppose I am solving it in Newton Raphson method.

For that,

$f(x=0) = -2.5 < 0$

$f(x=1) > 0$

So, there exists one root b/w $(0,1)$.

Let $x_0 = \frac{0+1}{2} = 0.5$

then $x_1 = x_0 - \frac{f(x)}{f'(x)}$

Iterate this step until $f(x) = 0$.

We can solve this in python as shown below. Then we get V_D as,

$$\boxed{V_D = 0.6258\,V}$$

So, $I_D = 10^{-14}\left(e^{\frac{0.6258}{0.0259}} - 1\right)$

$= 3.115 \times 10^{10} \times 10^{-14}\,A$

$$\therefore \boxed{I_D = 0.3115\,mA}$$

Python code for soliving the ablove equation:

Toplevel ▾ Fri 09:49

ex.py - /home/sivani/ex.py (3.6.9)

File Edit Format Run Options Window Help

```
import numpy as np
import math
import matplotlib.pyplot as plt

def func( x ):
    return (4*pow(10,-11)*(np.exp(x/0.0259)-1))+2*x-2.5

# Derivative of the above function
# which is (4*pow(10,-11)*(np.exp(x/0.0259)))/(0.0259)+2
def derivFunc( x ):
    return (4*pow(10,-11)*(np.exp(x/0.0259)))/(0.0259)+2

# Function to find the root
def newtonRaphson( x ):
    h = func(x) / derivFunc(x)
    while abs(h) >= 0.0001:
        h = func(x)/derivFunc(x)

        # x(i+1) = x(i) - f(x) / f'(x)
        x = x - h

    print("The value of the root is : ",
                                 "%.4f"% x)

x0=0.5
newtonRaphson(x0)
```



```
Python 3.6.9 Shell
File Edit Shell Debug Options Window Help
Python 3.6.9 (default, Oct  8 2020, 12:12:24)
[GCC 8.4.0] on linux
Type "help", "copyright", "credits" or "license()" for more information.
>>>
======================= RESTART: /home/sivani/ex.py =======================
The value of the root is :  0.6258
>>>
```

So, From analytical calculations Id=0.3115mA and Vd=0.6258 V.

2) c) Spice simulations for (a) and (b):

For (a):



File: /home/sivani/A1/Q2a.net Page 1 of 1

```
*Title:Solving for Id

*netlist
V1 1 0 2.5v
V2 2 3 0.7v
V3 4 0 0.7v
R1 1 2 2k
R2 3 4 2k

*control
.control
run
op
print -i(V1)
.endc
.end
```



```
sivani@sivani-Inspiron-5558: ~/A1
File Edit View Search Terminal Help
ngspice 1 -> source Q2a.net

Circuit: *title:solving for id

Doing analysis at TEMP = 27.000000 and TNOM = 27.000000

Doing analysis at TEMP = 27.000000 and TNOM = 27.000000


No. of Data Rows : 1
-i(v1) = 2.750000e-04
ngspice 1 ->
```

From spice simulation,we got Id = 0.275 mA (which is similar to our analytical value)

For (b):

File: /home/sivani/A1/Q2b.net Page 1 of 1

```
*Title:Q2(b)-Solving Id and Vd

*Netlist
V1 1 0 2.5V
R1 1 2 2k
D1 2 3 mydiode
R2 3 4 2k
D2 4 0 mydiode
.MODEL mydiode D (IS=1e-14)

*control
.control
run
op
print -i(V1)
print V(4)
.endc
.end
```

sivani@sivani-Inspiron-5558: ~/A1

File Edit View Search Terminal Help

```
ngspice 1 -> source Q2b.net

Circuit: *title:q2(b)-solving id and vd

Doing analysis at TEMP = 27.000000 and TNOM = 27.000000

Doing analysis at TEMP = 27.000000 and TNOM = 27.000000


No. of Data Rows : 1
-i(v1) = 3.124924e-04
v(4) = 6.250153e-01
ngspice 1 ->
```

From spice simulation,we got Id = 0.312 mA and Vd = 0.625 V(which are similar to our analytical values)

3) Simulation results for output response and Gain of circuit are-

<u>Netlist file:</u>

File: /home/sivani/A1/Q3.net Page 1 of 1

```
*Title:Gain of the circuit

*Netlist
Vin 0 in ac SIN(0V 0.5V 1kHz 0 0 0)
Rin in 1   1k
Rf  1  out 3k
R   out 0  1
G1  0 out (0,1) 1e6


*Analysis for 5ms,step size 0.01ms

.tran 0.01ms 5ms

.control
run

meas TRAN Voutmax MAX V(out)
meas TRAN Vinmax MAX V(in)
let Gain = -Voutmax/Vinmax
plot V(out) V(in)
print Gain
.endc
.end
```

Page 1 of 1

Output response:



The red wave is output response and the blue one is input voltage.

Gain of the circuit:



Here, Gain of the circuit is “-3”.

4) Mosfet Characteristics

a) ‘Id’ Vs ‘Vds’

<u>For Short Channel nmos :</u>

Netlist file:

```
*Title:Q4a- NMOS

.include TSMC180.lib
.model nch_tt nmos

*Netlist

*M1 ds gs 0 0 nch_tt W=15u L=10u
M1 ds gs 0 0 nch_tt W=0.27u L=0.18u
V1 ds 0 DC 0
V2 gs 0 DC 1.5V

.control
dc V1 0 1.8 0.01

*plot -i(V1) vs V(ds)
wrdata nmoss2.dat -i(v1) vs v(ds)
print -i(V1) V(ds)
.endc
.end
```



<u>For Long Channel nmos:</u>

Netlist file:

```
*Title:Q4a- NMOS

.include TSMC180.lib
.model nch_tt nmos

*Netlist

M1 ds gs 0 0 nch_tt W=15u L=10u

V1 ds 0 DC 0
V2 gs 0 DC 1.5V

.control
dc V1 0 1.8 0.01

*plot -i(V1) vs V(ds)
wrdata nmosl4.dat -i(v1) vs v(ds)
print -i(V1) V(ds)
.endc
.end
```



<u>For Short Channel pmos :</u>

Netlist file:



```
*Title:Q4a- PMOS

.include TSMC180.lib
.model pch_tt pmos

*Netlist

*M1 ds gs 0 0 pch_tt W=15u L=10u
M1 ds gs 0 0 pch_tt W=0.27u L=0.18u
V1 ds 0 DC 0
V2 gs 0 DC -1V

.control
dc V1 0 -1.8 -0.01

*plot -i(V1) vs v(ds)
wrdata pmoss4.dat -i(v1) vs v(ds)
.endc
.end
```

Plain Text Tab Width: 8 Ln 11, Col 14 INS



<u>For Long Channel pmos :</u>

Netlist file:

```
*Title:Q4a- PMOS

.include TSMC180.lib
.model pch_tt pmos

*Netlist

M1 ds gs 0 0 pch_tt W=15u L=10u

V1 ds 0 DC 0
V2 gs 0 DC -1V

.control
dc V1 0 -1.8 -0.01

*plot -i(V1) vs V(ds)
wrdata pmosl4.dat -i(v1) vs v(ds)
.endc
.end
```



4) b) Linear and Saturation regions

These regions are identified from ‘Id’ Vs ‘Vgs-Vth’ plot.

We use netlist files for simulating the ‘Id’ vs ‘Vgs’ plot for MOS devices.(just like above (a)).

For Long Channel nmos:

Netlist file:

```
*Title:Q4b- NMOS

.include TSMC180.lib
.model nch_tt nmos

*Netlist

M1 ds gs 0 0 nch_tt W=15u L=10u

V1 ds 0 DC 1.8
V2 gs 0 DC 0

.control
dc V2 0 1.8 0.01

*plot -i(V1) vs V(gs)
wrdata plot1.dat -i(v1) vs v(gs)
.endc
.end
```



Fig-1

For Short Channel nmos:

Netlist file:

```
*Title:Q4b- NMOS

.include TSMC180.lib
.model nch_tt nmos

*Netlist

M1 ds gs 0 0 nch_tt W=0.27u L=0.18u
V1 ds 0 DC 1.8
V2 gs 0 DC 0

.control
dc V2 0 1.8 0.01

*plot -i(V1) vs V(gs)
wrdata plot2.dat -i(v1) vs v(gs)
.endc
.end
```



Fig-2

<u>For Short Channel pmos:</u>

Netlist file:

```
*Title:Q4b- PMOS

.include TSMC180.lib
.model pch_tt pmos

*Netlist

M1 ds gs 0 0 pch_tt W=0.27u L=0.18u
V1 ds 0 DC -1.8
V2 gs 0 DC 0

.control
dc V2 0 -1.8 -0.01

*plot -i(V1) vs V(gs)
wrdata plot3.dat -i(v1) vs v(gs)
.endc
.end
```



Fig – 3

<u>For Long Channel pmos:</u>

Netlist file:

```
*Title:Q4b- PMOS

.include TSMC180.lib
.model pch_tt pmos

*Netlist

M1 ds gs 0 0 pch_tt W=15u L=10u
V1 ds 0 DC -1.8
V2 gs 0 DC 0

.control
dc V2 0 -1.8 -0.01

*plot -i(V1) vs V(gs)
wrdata plot4.dat -i(v1) vs v(gs)
.endc
.end
```



Fig-4

In all these above figures -1,2,3 and 4 :

- The region to the right of 'Id' Vs 'Vgs-Vth' is Saturation region .
- Where as the region to the left of 'Id' Vs 'Vgs-Vth' is Linear region.

4) c) Small signal output resistance is given by inverse of slope of Id Vs Vds curve in saturation region.

Slope can be calculated by taking two points present in saturation region in the plot.

For example,

Lets take slope of Id Vs Vds plot in short channel nmos when Vgs=1.5 V [Take two points in saturation region from simulation attached below]

Slope = ΔI / ΔV = (0.00115x10^-4) / (10^-2)

=115 x 10^-7

So, small signal output resistance = 1 / (slope) = 86956.5 Ω



Similarly, small signal output resistance in other three cases(i.e., Long channel nmos,short and long channel pmos) are calculated.

They are obtained as-(When Vgs = 1.5 V)

- In long channel nmos,

  slope = 7.104X10^-7

  So,Small signal output resistance = 1.4 MΩ
- In short channel pmos,

  slope = 2.81X10^-6

  So,Small signal output resistance = 355871 Ω

- In long channel pmos,

  slope = 2.127X10^-7

  So,Small signal output resistance = 4.7 MΩ

4) d) Id – Vg characteristics on a log-linear scale:

For short channel nmos:

Netlist file:



```
*Title:Short channel NMOS
.include TSMC180.lib
.model nch_tt nmos

*Netlist

*M1 ds gs 0 0 nch_tt W=15u L=10u
M1 ds gs 0 0 nch_tt W=0.27u L=0.18u
V1 ds 0 DC 1.8
V2 gs 0 DC 0

.control
dc V2 0 1.8 0.01

plot log10(-i(V1)) vs V(gs)


.endc
.end
```



Fig -1: Id-Vg characteristics of short channel nmos

Subthreshold slope:It is defined as the slope of Id-Vg curve in the region until Vg reaches Vt .

Considering two points in this region (110,-9.5) and (240,-8)

Sub threshold slope = ΔV/ΔI=130/1.5



So,Sub threshold slope = 86.67 mV/decade

For Long channel nmos:

Netlist file:

```
*Title:Long channel NMOS
.include TSMC180.lib
.model nch_tt nmos

*Netlist

M1 ds gs 0 0 nch_tt W=15u L=10u
*M1 ds gs 0 0 nch_tt W=0.27u L=0.18u
V1 ds 0 DC 1.8
V2 gs 0 DC 0

.control
dc V2 0 1.8 0.01

plot log10(-i(V1)) vs V(gs)

.endc
.end
```



Fig -2: Id-Vg characteristics of long channel nmos

Considering two points in this region (70,-10) and (140,-9.2)

Sub threshold slope = ΔV/ΔI

= 70 /0.8

= 87.5 mV/decade



So,subthreshold slope is 87.5 mV/decade.

For short channel pmos:

Netlist file:



```
*Title:Short channel PMOS
.include TSMC180.lib
.model pch_tt pmos

*Netlist

*M1 ds gs 0 0 pch_tt W=15u L=10u
M1 ds gs 0 0 pch_tt W=0.27u L=0.18u
V1 ds 0 DC -1.8
V2 gs 0 DC 0

.control
dc V2 0 -1.8 -0.01

plot log10(abs(-i(V1))) vs V(gs)

.endc
.end
```



Fig -3: Id-Vg characteristics of short channel pmos

Considering two points in this region (-210,-8.8) and (-190,-9) then

Sub threshold slope = ΔV/ΔI

= 20 /-0.2

= -100 mV/decade



So,sub threshold slope is -100 mV/dec.

<u>For long channel pmos:</u>

Netlist file:

```
*Title:Long channel PMOS
.include TSMC180.lib
.model pch_tt pmos

*Netlist

M1 ds gs 0 0 pch_tt W=15u L=10u
*M1 ds gs 0 0 pch_tt W=0.27u L=0.18u
V1 ds 0 DC -1.8
V2 gs 0 DC 0

.control
dc V2 0 -1.8 -0.01

plot log10(abs(-i(V1))) vs V(gs)

.endc
.end
```



Fig -4: Id-Vg characteristics of long channel pmos

Considering two points in this region
(-220,-8.8) and (-210,-8.9) then
Sub threshold slope = ΔV/ΔI

= 10 /-0.1

= -100 mV/decade

So,sub threshold slope is -100 mV/dec.



5) Propagation Delay

Given first order RC circuit having proparation delay for an ideal step input(tr,in = 0) is tp = 0.69RC. Further, the output rise time tr = 2.2RC.

To verify this circuit satisfying given relation I am choosing R and C values as R = 1k Ω and C = 1pF then

Propagation delay (tp) should become 0.69 ns and Rise time should become 2.2 ns.

5) a)So,For this we shall verify simulating the circuit in ngspice.

Netlist file:

```
**Propagation delay*
V1 in 0 PULSE(0 1 0 0 0)
R1 in out 1k
C1 out 0 1p
.tran 0.01ns 15n

.control
run
plot V(out) V(in)
meas tran peak MAX V(out)
let vp = peak/2
meas tran Tp find time when V(out)=vp cross=1
print Tp
let v2 = 0.1*peak
let v3 = 0.9*peak
meas tran T2 find time when V(out)=v2
meas tran T3 find time when V(out)=v3
let tr= T3-T2
print tr
.endc
.end
```



Output and Input wave forms

```
sivani@sivani-Inspiron-5558: ~/A1
File  Edit  View  Search  Terminal  Help

Doing analysis at TEMP = 27.000000 and TNOM = 27.000000

Warning: v1: no DC value, transient time 0 value used

Initial Transient Solution
--------------------------

Node                                   Voltage
----                                   -------
in                                           0
out                                          0
v1#branch                                    0

 Reference value :  0.00000e+00

No. of Data Rows : 1512
peak                 =  9.999997e-01 at=  1.500000e-08
tp                   =  6.981516e-10
tp = 6.981516e-10
t2                   =  1.103673e-10
t3                   =  2.307575e-09
tr = 2.197208e-09
ngspice 1 ->
```

Simulation Result

From simulation results, tp = 0.6901 ns and tr = 2.19 ns .

So,We can observe that both the results are same and accurate by choosing those R and C values.

5) b) Considering a non-ideal step input with 10 ps < tr,in < 10 ns and simulating the circuit

Netlist file:

- Varying rise time in input signal and do simulation.

```
**Propagation delay*
V1 in 0 PULSE(0 1 0 3e-9 0) ** changing tr in the given range
R1 in out 1k
C1 out 0 1p
.tran 0.01ns 15n

.control
run
*plot V(out) V(in)
meas tran peak MAX V(out)
let vp = peak/2
meas tran Tp find time when V(out)=vp cross=1
print Tp
let v2 = 0.1*peak
let v3 = 0.9*peak
meas tran T2 find time when V(out)=v2
meas tran T3 find time when V(out)=v3
let tr= T3-T2
print tr
wrdata p1.data Tp tr
.endc
.end
```

Saving file "/home/siva...   Plain Text ▾   Tab Width: 8 ▾   Ln 2, Col 62   ▾   INS

Noting the points Tp and Tr and plotting in python yields a plot as shown in Fig -1.

vel ▾   Sat 18:43

p1.py - /home/sivani/A1/p1.py (3.6.9)

File Edit Format Run Options Window Help

```python
import numpy as np
import matplotlib.pyplot as plt


p1=[2.197207e-09,7.435717e-10]
p2=[2.197207e-09,7.948166e-10]
p3=[2.200978e-09,8.469045e-10]
p4=[2.229168e-09,9.535511e-10]
p5=[2.249937e-09,1.008116e-09]
p6=[2.273902e-09,1.063490e-09]
p7=[2.300584e-09,1.119688e-09]
p8=[2.329607e-09,1.176682e-09]
p9=[2.360715e-09,1.234478e-09]
p10=[2.757257e-09,1.841406e-09]
p11=[3.264004e-09,2.410201e-09]
p12=[3.847679e-09,2.947523e-09]
p13=[4.488023e-09,3.468824e-09]
p14=[5.170686e-09,3.981278e-09]
p15=[6.624284e-09,4.992757e-09]
p16=[7.381933e-09,5.494652e-09]
p17=[8.153285e-09,5.994138e-09]
p18=[2.197208e-09, 7.334220e-10]

x_val=[p18[0],p1[0],p2[0],p3[0],p4[0],p5[0],p6[0],p7[0],p8[0],p9[0],p10[0],p11[0
y_val=[p18[1],p1[1],p2[1],p3[1],p4[1],p5[1],p6[1],p7[1],p8[1],p9[1],p10[1],p11[1
plt.plot(x_val,y_val)
plt.grid()
plt.xlabel("Tr")
plt.ylabel("Tp")
plt.title("Plot of propagation delay(Tp) aganist Rise time(Tr)")
plt.show()
```



Fig – 1

5) c)

5) c) considering a non-ideal step input as,



So, $v_{in}(t) = \begin{cases} t/\alpha, & t \le \alpha \\ 1, & t > \alpha \end{cases}$

$$\frac{dv_{in}(t)}{dt} = \begin{cases} 1/\alpha, & t < \alpha \\ 0, & t > \alpha \end{cases}$$

$$\therefore \frac{dv_{in}(t)}{dt} = \frac{d}{dt}\left(IR + \frac{\int I\,dt}{c}\right)$$

$$\therefore \frac{dv_{in}}{dt} = \frac{I}{c} + R \cdot \frac{dI}{dt}$$

For $t<\alpha$,

$$\frac{1}{\alpha} = \frac{I}{c} + R \cdot \frac{dI}{dt}$$

$$\Rightarrow I - \frac{c}{\alpha} = -RC\frac{dI}{dt}$$

$$\therefore \frac{dt}{-RC} = \frac{dI}{I - c/\alpha}$$

At $t=0$, $v_{in}=0 \Rightarrow I=0$

$$\int_{t=0}^{t=t} \frac{-dt}{RC} = \int_{I=0}^{I=I} \frac{dI}{I - c/\alpha}$$

$$\Rightarrow \frac{-t}{RC} = \ln\left(\frac{I - c/\alpha}{-c/\alpha}\right)$$

$$\therefore \boxed{I = \frac{c}{\alpha}\left(1 - e^{-t/RC}\right)}$$

For $t > \alpha$,

$$0 = \frac{I}{C} + R \cdot \frac{dI}{dt}$$

$$\int_{I_0}^{I} -\frac{dI}{I} = \int_{\alpha}^{t} \frac{1}{RC}\, dt$$

$$\cancel{\log}\left(I/I_0\right) = e^{-t/RC} \cdot e^{\alpha/RC}$$

At $t = \alpha$.

$$\boxed{I = I_0 = \frac{c}{\alpha}\left(1 - e^{-\alpha/RC}\right)}$$

For V_{in}, $t_r = 0.8\alpha$ as it linearly taking time 'α' to raise from '0' to '1'. So, time taking from 10% to 90% is $0.9\alpha - 0.1\alpha = 0.8\alpha$

& $t_p = 0.5\alpha$



Solving for V_{out},

$$V_{in} - \frac{dV_0}{dt} RC - V_0 = 0$$

$$\frac{dV_0}{dt} = \frac{V_{in}}{RC} + \frac{V_0}{RC}$$

$$\frac{dv_0}{dt} = \frac{1}{\alpha}\frac{t}{RC} + \frac{v_0}{RC} \quad (\text{for } t \le \alpha)$$

For case-1:-

Solving this equation we get,

$$\boxed{v_{out} = v_0 = \frac{t - RC(1 - e^{-t/RC})}{\alpha} \quad (t \le \alpha)}$$

~~For $t \ge \alpha$, $v_{in} = 1$~~

$$\cancel{\frac{dv_0}{dt} = \frac{1}{RC} + \frac{v_0}{RC}}$$

For case-2:-

$$v_{out} = v_0 = 1 - \frac{RC}{\alpha}(1 - e^{-\alpha/RC})(e^{-\frac{(t-\alpha)}{RC}})$$

So, Propagation delay in both cases will be as -

In case-1: $t_p = RC\ln\left(\frac{2RC}{\alpha}(e^{\alpha/RC} - 1)\right)$

In case-2: $t_p = \underbrace{\frac{\alpha}{2} + RC(1 - e^{-t_p/RC})}$

This is an implicit function so can't be solved just we can say that it is increasing function